

Datetime display formats — Display formats for dates and times[Description](#)[Quick start](#)[Syntax](#)[Remarks and examples](#)[Also see](#)

Description

Stata stores dates and times numerically in one of eight units. The value of a Stata date might be 18,282 or even 1,579,619,730,000. Place the appropriate format on it, and the 18,282 is displayed as 20jan2010 (%td). The 1,579,619,730,000 is displayed as 20jan2010 15:15:30 (%tc).

If you specify additional format characters, you can change how the result is displayed. Rather than 20jan2010, you could change it to 2010.01.20; January 20, 2010; or 1/20/10. Rather than 20jan2010 15:15:30, you could change it to 2010.01.20 15:15; January 20, 2010 3:15 pm; or Wed Jan 20 15:15:30 2010.

See [\[D\] Datetime](#) for an introduction to Stata's dates and times.

Quick start

Format daily dates stored in `datevar` to display as 15mar2005

```
format datevar %td
```

Format daily dates stored in `datevar` to display as 3/15/05

```
format datevar %tdnn/DD/YY
```

Format daily dates stored in `datevar` to display as Tue Mar. 15

```
format datevar %tdDay_Mon._DD
```

Format dates and times stored in `timevar` to display as 15mar2005 14:30:00

```
format timevar %tc
```

Format dates and times stored in `timevar` to display as 14:30

```
format timevar %tcHH:MM
```

Format dates and times stored in `timevar` to display as 2:30 PM

```
format timevar %tchh:mm_AM
```

Syntax

The formats for displaying Stata dates and times are

Stata date type	Display format
datetime/c	%tc [<i>details</i>]
datetime/C	%tC [<i>details</i>]
date	%td [<i>details</i>]
weekly date	%tw [<i>details</i>]
monthly date	%tm [<i>details</i>]
quarterly date	%tq [<i>details</i>]
half-yearly date	%th [<i>details</i>]
yearly date	%ty [<i>details</i>]

The optional *details* allows you to control how results appear and is composed of a sequence of the following codes:

Code	Meaning	Output
CC	century-1	01–99
cc	century-1	1–99
YY	2-digit year	00–99
yy	2-digit year	0–99
JJJ	day within year	001–366
jjj	day within year	1–366
Mon	month	Jan, Feb, . . . , Dec
Month	month	January, February, . . . , December
mon	month	jan, feb, . . . , dec
month	month	january, february, . . . , december
NN	month	01–12
nn	month	1–12
DD	day within month	01–31
dd	day within month	1–31
DAYNAME	day of week	Sunday, Monday, . . . (aligned)
Dayname	day of week	Sunday, Monday, . . . (unaligned)
Day	day of week	Sun, Mon, . . .
Da	day of week	Su, Mo, . . .
day	day of week	sun, mon, . . .
da	day of week	su, mo, . . .

h	half	1–2
q	quarter	1–4
WW	week	01–52
ww	week	1–52
HH	hour	00–23
Hh	hour	00–12
hH	hour	0–23
hh	hour	0–12
MM	minute	00–59
mm	minute	0–59
SS	second	00–60 (sic, due to leap seconds)
ss	second	0–60 (sic, due to leap seconds)
.s	tenths	.0–.9
.ss	hundredths	.00–.99
.sss	thousandths	.000–.999
am	show am or pm	am or pm
a.m.	show a.m. or p.m.	a.m. or p.m.
AM	show AM or PM	AM or PM
A.M.	show A.M. or P.M.	A.M. or P.M.
.	display period	.
,	display comma	,
:	display colon	:
-	display hyphen	-
	display space	
/	display slash	/
\	display backslash	\
!c	display character	c
+	separator (see note)	

Note: + displays nothing; it may be used to separate one code from the next to make the format more readable. + is never necessary. For instance, %tchh:MM+am and %tchh:MMam have the same meaning, as does %tc+hh+:+MM+am.

When *details* is not specified, it is equivalent to specifying

Format	Implied (fully specified) format
<code>%tC</code>	<code>%tCDDmonCCYY_HH:MM:SS</code>
<code>%tc</code>	<code>%tcDDmonCCYY_HH:MM:SS</code>
<code>%td</code>	<code>%tdDDmonCCYY</code>
<code>%tw</code>	<code>%twCCYY!www</code>
<code>%tm</code>	<code>%tmCCYY!mnn</code>
<code>%tq</code>	<code>%tqCCYY!qq</code>
<code>%th</code>	<code>%thCCYY!hh</code>
<code>%ty</code>	<code>%tyCCYY</code>

That is, typing

```
. format mytimevar %tc
```

has the same effect as typing

```
. format mytimevar %tcDDmonCCYY_HH:MM:SS
```

Format `%tcDDmonCCYY_HH:MM:SS` is interpreted as

%	t	c	DDmonCCYY_HH:MM:SS
all formats start with %	it is a datetime format	variable coded in milliseconds	formatting codes specify how to display value

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

Specifying display formats

Times are truncated, not rounded, when displayed

Specifying display formats

Rather than using the default format `20jan2010`, you could display the daily date in one of these formats:

```
2010.01.20
January 20, 2010
1/20/10
```

Likewise, rather than displaying the `datetime/c` variable in the default format `20jan2010 15:15:30`, you could display it in one of these formats:

```
2010.01.20 15:15
January 20, 2010 3:15 pm
Wed Jan 20 15:15:30 2010
```

Here is how to do it:

1. 2010.01.20
format *mytdvar* %tdCCYY.NN.DD
2. January 20, 2010
format *mytdvar* %tdMonth_dd,_CCYY
3. 1/20/10
format *mytdvar* %tdnn/dd/YY
4. 2010.01.20 15:15
format *mytcvar* %tcCCYY.NN.DD_HH:MM
5. January 20, 2010 3:15 pm
format *mytcvar* %tcMonth_dd,_CCYY_hh:MM_am
Code am at the end indicates that am or pm should be displayed, as appropriate.
6. Wed Jan 20 15:15:30 2010
format *mytcvar* %tcDay_Mon_DD_HH:MM:SS_CCYY

In examples 1 to 3, the formats each begin with %td, and in examples 4 to 6, the formats begin with %tc. It is important that you specify the opening correctly—namely, as % + t + *third_character*. The third character indicates the particular encoding type, which is to say, how the numeric value is to be interpreted. You specify %tc... for datetime/c variables, %tC... for datetime/C, %td... for date, and so on.

The default format for datetime/c and datetime/C variables omits the fraction of seconds; 15:15:30.000 is displayed as 15:15:30. If you wish to see the fractional seconds, specify the format

```
%tCDDmonCCYY_HH:MM:SS.sss
```

or

```
%tCDDmonCCYY_HH:MM:SS.sss
```

as appropriate.

Times are truncated, not rounded, when displayed

Consider the time 11:32:59.999. Other, less precise, ways of writing that time are

```
11:32:59.99
11:32:59.9
11:32:59
11:32
```

That is, when you suppress the display of more-detailed components of the time, the parts that are displayed are not rounded. Stata displays time just as a digital clock would; the time is 11:32 right up until the instant that it becomes 11:33.

Also see

[D] **Datetime** — Date and time values and variables

[D] **Datetime business calendars** — Business calendars

[D] **Datetime conversion** — Converting strings to Stata dates

[D] **Datetime durations** — Obtaining and working with durations

[D] **Datetime relative dates** — Obtaining dates and date information from other dates

[D] **Datetime values from other software** — Date and time conversion from other software